

# Symmetry computations in Maple

Author: Dennis The

Date: 26 January 2021

This worksheet is a supplement to the lecture "Classifying homogeneous geometric structures (Lecture 1)" (<http://cft.edu.pl/grieg/activities.html>).

In this worksheet, we give several examples of how to use Maple to compute symmetries of various differential geometric structures.

```
> restart: with(DifferentialGeometry): with(GroupActions): with
(LieAlgebras): with(JetCalculus):
sym:=v->InfinitesimalSymmetriesOfGeometricObjectFields(v,output=
"list");
```

## Metrics

```
M > DGsetup([x,y],M);
```

*Manifold: M* (1.1)

```
M > g:=evalDG(dx &s dx + dy &s dy);
```

$g := dx \otimes dx + dy \otimes dy$  (1.2)

```
M > S:=sym(g);
nops(%);
```

$S := \left[ -y \partial_x + x \partial_y, \partial_y, \partial_x \right]$   
3 (1.3)

## Conformal structures

```
M > DGsetup([x,y,z,w],M);
```

*Manifold: M* (2.1)

```
M > c:=[evalDG(dy &s dy + dz &s dz + dw &s dx + y^2*dw &s dw)];
```

$c := \left[ \frac{1}{2} dx \otimes dw + dy \otimes dy + dz \otimes dz + \frac{1}{2} dw \otimes dx + y^2 dw \otimes dw \right]$  (2.2)

```
M > S:=sym([c]);
nops(%);
```

$S := \left[ x \partial_x + \frac{y}{2} \partial_y + \frac{z}{2} \partial_z, \partial_w, -2z \partial_x + w \partial_z, 2e^{-w} y \partial_x + e^{-w} \partial_y, -2e^w y \partial_x + e^w \partial_y, \partial_x \right]$   
7 (2.3)

```
M > LieAlgebraData(S,alg):
DGsetup(%);
MultiplicationTable();
```

*Lie algebra: alg*

alg	$e1$	$e2$	$e3$	$e4$	$e5$	$e6$	$e7$
$e1$	0	0	$-\frac{1}{2} e3$	$-\frac{1}{2} e4$	$-\frac{1}{2} e5$	$-\frac{1}{2} e6$	$-e7$
$e2$	0	0	$e4$	0	$-e5$	$e6$	0
$e3$	$\frac{1}{2} e3$	$-e4$	0	$2 e7$	0	0	0
$e4$	$\frac{1}{2} e4$	0	$-2 e7$	0	0	0	0
$e5$	$\frac{1}{2} e5$	$e5$	0	0	0	$-4 e7$	0
$e6$	$\frac{1}{2} e6$	$-e6$	0	0	$4 e7$	0	0
$e7$	$e7$	0	0	0	0	0	0

(2.4)

```
alg > Series("Derived");
Series("Lower");
```

$$\left[ [e1, e2, e3, e4, e5, e6, e7], \left[ -\frac{1}{2} e3, -\frac{1}{2} e4, -\frac{1}{2} e5, -\frac{1}{2} e6, -e7 \right], \left[ \frac{1}{2} e7 \right], [ ] \right]$$

$$\left[ [e1, e2, e3, e4, e5, e6, e7], \left[ -\frac{1}{2} e3, -\frac{1}{2} e4, -\frac{1}{2} e5, -\frac{1}{2} e6, -e7 \right], \left[ -\frac{1}{4} e3, \right. \right. \quad (2.5)$$

$$\left. \left. \frac{1}{2} e4, -e7, -\frac{1}{4} e5, -\frac{1}{4} e6 \right] \right]$$

```
alg > Nilradical();
Radical();
```

$$[e3, e4, e5, e6, e7]$$

$$[e7, e6, e5, e4, e3, e2, e1] \quad (2.6)$$

```
alg > Center();
```

$$[ ] \quad (2.7)$$

## 2nd order ODE

```
M > DGsetup([x,y,p],M);
```

$$\text{Manifold: } M \quad (3.1)$$

```
M > f:=exp(p):
E:=evalDG([D_x+p*D_y+f*D_p]);
V:=evalDG([D_p]);
ODE:=[E,V];
```

$$E := \left[ \partial_x + p \partial_y + e^p \partial_p \right]$$

$$V := \left[ \partial_p \right]$$

$$ODE := \left[ \left[ \partial_x + p \partial_y + e^p \partial_p \right], \left[ \partial_p \right] \right] \quad (3.2)$$

```
M > S:=sym(ODE);
```

```
nops(%);
```

$$S := \left[ -x \partial_x + (x-y) \partial_y + \partial_p, \partial_y, \partial_x \right]$$

3

(3.3)

```
M > LieAlgebraData(S,alg):
DGsetup(%);
MultiplicationTable();
```

*Lie algebra: alg*

alg	e1	e2	e3
e1	0	e2	-e2 + e3
e2	-e2	0	0
e3	e2 - e3	0	0

(3.4)

```
M > DGsetup([x],[y],J,2);
```

```
X:=evalDG([D_x,D_y[],x*D_x+(y[]-x)*D_y[]]);
```

```
# These were the point symmetries given above.
# (Rather, their projections to (x,y)-space.)
# We've written them here on the zeroth jet space J^0(R,R).
```

*Jet Bundle: J*

$$X := \left[ \partial_x, \partial_{y[]}, x \partial_x - (-y[] + x) \partial_{y[]} \right]$$

(3.5)

```
J > # Let's prolong the symmetries up to the second jet space
J^2(R,R).
```

```
X2:=map(v->Prolong(v,2),X);
```

```
# These prolonged point symmetries are tangent vector fields
to y[1,1]=exp(y[1]). Let's check:
```

```
[seq(LieDerivative(X2[i],y[1,1]-exp(y[1])),i=1..3)];
eval(%,y[1,1]=exp(y[1]));
```

$$X2 := \left[ \partial_x, \partial_{y[]}, x \partial_x - (-y[] + x) \partial_{y[]} - \partial_{y[1]} - y_{1,1} \partial_{y[1,1]} \right]$$

$$\left[ 0, 0, e^{y_1} - y_{1,1} \right]$$

$$[0, 0, 0]$$

(3.6)

## (2,3,5)-distributions

```
M > DGsetup([x,y,p,q,z],M);
```

*Manifold: M*

(4.1)

```
M > # Hilbert-Cartan
```

```
f:=q^2;
dist:=evalDG([D_x+p*D_y+q*D_p+f*D_z,D_q]);
```

```
LieBracket(dist[1],dist[2]);
LieBracket(dist[1],%);
LieBracket(dist[2],%%);
```

$$\begin{aligned}
dist := & \left[ \partial_x + p \partial_y + q \partial_p + q^2 \partial_z, \partial_q \right] \\
& - \partial_p - 2 q \partial_z \\
& \partial_y \\
& - 2 \partial_z
\end{aligned} \tag{4.2}$$

```

M > S:=sym([dist]);
nops(%);
pt:=[x=0,y=0,p=0,q=1,z=0]:
IsotropyFiltration(S,pt):
map(nops,%);

```

$$\begin{aligned}
S := & \left[ \left( -\frac{3 q y}{2} + p^2 \right) \partial_x + \left( -\frac{3}{2} y p q + \frac{3}{4} y z + \frac{2}{3} p^3 \right) \partial_y + \left( -\frac{3 q^2 y}{4} + \frac{3 p z}{4} \right) \partial_p - \right. \\
& \left( \frac{1}{2} p q^2 - \frac{3}{4} q z \right) \partial_q - \left( \frac{y q^3}{2} - \frac{3 z^2}{4} \right) \partial_z, \frac{x^2}{4} \partial_y + \frac{x}{2} \partial_p + \frac{1}{2} \partial_q + p \partial_z, \partial_z, x \partial_y \\
& + \partial_p, \partial_y, \partial_x, - \left( -3 y x + 2 p x^2 - \frac{1}{2} q x^3 \right) \partial_x - \left( -3 y^2 - \frac{1}{2} p q x^3 + p^2 x^2 + \frac{1}{4} x^3 z \right. \\
& \left. \right) \partial_y - \left( -3 y p - \frac{1}{4} q^2 x^3 + p^2 x + \frac{3}{4} x^2 z \right) \partial_p + \left( \frac{1}{2} q^2 x^2 - q p x + 2 p^2 - \frac{3}{2} x z \right) \partial_q \\
& + \left( 3 y z + \frac{1}{6} q^3 x^3 - 3 p x z + \frac{4}{3} p^3 \right) \partial_z, - \left( -\frac{3}{2} q x^2 + 4 p x - 3 y \right) \partial_x - \left( \right. \\
& - \frac{3}{2} x^2 p q + 2 p^2 x + \frac{3}{4} x^2 z \left. \right) \partial_y - \left( -\frac{3}{4} q^2 x^2 + p^2 + \frac{3}{2} x z \right) \partial_p - \left( -q^2 x + p q \right. \\
& + \frac{3}{2} z \left. \right) \partial_q - \left( -\frac{q^3 x^2}{2} + 3 p z \right) \partial_z, \left( 2 p - \frac{3 q x}{2} \right) \partial_x + \left( -\frac{3}{2} q p x + p^2 + \frac{3}{4} x z \right) \partial_y \\
& - \left( \frac{3 q^2 x}{4} - \frac{3 z}{4} \right) \partial_p - \frac{q^2}{2} \partial_q - \frac{q^3 x}{2} \partial_z, x^2 \partial_x + 3 y x \partial_y + (p x + 3 y) \partial_p + (-q x \\
& + 4 p) \partial_q + 4 p^2 \partial_z, -y \partial_y - p \partial_p - q \partial_q - 2 z \partial_z, -q \partial_x - \left( p q - \frac{z}{2} \right) \partial_y - \frac{q^2}{2} \partial_p \\
& - \frac{q^3}{3} \partial_z, -\frac{x^3}{6} \partial_y - \frac{x^2}{2} \partial_p - x \partial_q - \left( 2 p x - 2 y \right) \partial_z, -x \partial_x - 2 y \partial_y - p \partial_p \\
& \left. - z \partial_z \right]
\end{aligned}$$

14

[9, 2, 0] (4.3)

```

M > # f=exp(q)

```

```

f:=exp(q):
dist:=evalDG([D_x+p*D_y+q*D_p+f*D_z,D_q]);
S:=sym([dist]);
nops(%);
pt:=[x=0,y=0,p=0,q=0,z=0]:
IsotropyFiltration(S,pt):

```

$$\begin{matrix} 7 \\ [2, 0] \end{matrix} \quad (4.4)$$